



Núcleo de Economia Regional e Urbana da Universidade de São Paulo



Interregional General Equilibrium Modeling

Eduardo Haddad

Outline

✓ What is the CEER Model?

Overview

Building blocks

A multi-sectoral, multi-regional bottom-up CGE model of Colombia's 32 Departments and the capital city, Bogotá

- each region is modeled as an economy in its own right
- region-specific prices
- region-specific industries
- region-specific consumers

Based on the comparative-static B-MARIA and MMRF models

Database makes allowance for interregional, intra-regional and international trade

 explicit representations of regional and federal government financial accounts

Stylized flows



Embedded SAM

		REGION 1					REGION R											
		FACTORS	AGENTS H G	SECTORS	LOCAL PRODUCTS	COMPOSITE PRODUCTS (SUPPLY)	FACTORS	AGEN H	TS G	SECTORS	LOCAL PRODUCTS	COMPOSITE PRODUCTS (SUPPLY)	FEDERAL GOV.	IMPORTS	EXPORTS	ACCUM.	ROW	TOTAL
REGION 1	FACTORS			W ¹														W^1
	H AGENTS G	Wh ¹¹	Thrg ¹	\bigcirc			Wh ^{1R}						Thfg ¹					Yh^1
			IDrgh ¹		TIrg ¹								Tfg ¹					Yrg ¹
	SECTORS				PMI ¹	\frown									E^1			PCF ¹
	LOCAL PRODUCTS					Ppm ¹¹						Ppm ^{1R}						Ppm ¹
	COMPOSITE PRODUCTS (DEMAND)		Ch ¹ Crg ¹	CI1									Cfg^1			\mathbf{I}^1		D^1
REGION R	FACTORS									W^R			\frown					W ^R
	H	Wh ^{R1}					Wh ^{RR}		Thrg ^R			(Thfg ^R					Yh^{R}
	G	~						IDrgh ^R			TIrg ^R		Tfg ^R					Yrg ^R
	SECTORS										PMI ^R				E^{R}			PCF ¹
	LOCAL PRODUCTS					Ppm ^{R1}						Ppm ^{RR}						Ppm ^R
	COMPOSITE PRODUCTS (DEMAND)							Ch^{R}	Crg ^R	CI ^R			Cfg ^R			I^R		D^R
	FEDERAL GOVERNMENT		IDfgh ¹		TIfg ¹	\frown		IDfgh ^R			TIfg ^R			Tim	Tex			Yfg
	IMPORTS					Ipm ¹						Ipm ^R						Ipm
	EXPORTS																Epm	Epm
	ACCUMULATION		Sh ¹ Srg ¹					$\operatorname{Sh}^{\mathrm{R}}$	Srg ^R				Sfg				SBC	Ι
	ROW													М				Yrw
	FINANCIAL/ASSET ADJUSTMENT		FAh ¹ FArg ¹					FAh ^R	FArg ^R				FAfg					FA
	TOTAL	\mathbf{W}^1	Yh ¹ Yrg ¹	PCF ¹	Ppm ¹	O^1	W ^R	Yh ^R	Yrg ^R	PCF ^R	Ppm ^R	O ^R	Yfg	Ipm	Epm	Ι	Yrw	

The Role of Transportation Services in CEER



Core database

		ABSORPTION MATRIX										
		1	2	3	4	5	6					
		Producers	Investors	Household	Export	Regional Govt.	Federal Govt.					
	Size	J x Q	J x Q	Q	1	Q	Q					
Basic Flows	I x S	BAS1	BAS2	BAS3	BAS4	BAS5	BAS6					
Margins	I x S x R	MAR1	MAR2	MAR3	MAR4	MAR5	MAR6					
Taxes	I x S	TAX1	TAX2	TAX3	TAX4	TAX5	TAX6					
Labor	1	LABR					8					
Capital	1	CPTL	I = number of commodities									
Other	1	OCTS	OCTS J = number of industries									

R = number of commodities used as margins

Q = number of regions

S = Q domestic regions + 1 foreign import

Commodity flows are valued at "basic prices" (BAS):

do not include user-specific taxes or margins

For each user of each imported good and each domestic good, there are numbers showing:

- tax levied on that usage (TAX)
- usage of margins transport (MAR)

Single-production:

- each commodity may be produced by one industry
- each industry may produce one commodity

For each industry the total cost of production is equal to the total value of output

For each commodity the total value of sales is equal to the total value of output

Domestic producers

J industries in Q regions

Investors

J industries in Q regions

Households

one representative household for each of the Q regions

Each of the I commodity types can be obtained from the region, from other regions, or imported from overseas

Aggregate foreign purchaser of exports

Other demand category corresponding to the Q regional governments

Other demand category corresponding to the Central government in the Q regions

Commodity taxes and Margins explicitly recognised

Notation

Main User Numbers

- $1 \Leftrightarrow$ firms, current production,
- $2 \Leftrightarrow$ firms, capital creation;
- $3 \Leftrightarrow households$,
- 4 \Leftrightarrow foreign exports,
- 5 \Leftrightarrow regional government,
- $6 \Leftrightarrow$ Central government,

The number 0 is also used to denote basic prices and values.

Source dimensions

- a \Leftrightarrow all sources, i.e., 33 regional sources and 1 foreign;
- $r \Leftrightarrow$ regional sources only;
- t \Leftrightarrow two sources, i.e., a domestic composite source and foreign;
- $c \Leftrightarrow$ domestic composite source only;
- o \Leftrightarrow domestic-foreign composite source only.



What is the CEER Model?



Building blocks

Fipe - Fundação Instituto de Pesquisas Econômicas 12

Building blocks

✓ Producer's demands for inputs

- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Production nest



Demand for intermediate inputs, other costs and prices

```
E_x1a # Demand for goods by all sources, User 1 #
(all,i,COM) (all,s,ALLSOURCE) (all,j,IND) (all,q,REGDEST)
x1a(i,s,j,q)=IS_DOM(s)*(x1c(i,j,q)-
SIGMA1C(i)*(p1a(i,s,j,q)-p1c(i,j,q)))
+IS_IMP(s)*(x1o(i,j,q)-SIGMA10(i)*(p1a(i,"foreign",j,q)-
p1o(i,j,q)));
```

```
E_plo # Price of domestic/foreign composite, User 1 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
(TINY+PVAL10(i,j,q))*plo(i,j,q) =
sum(s,ALLSOURCE,PVAL1A(i,s,j,q)*pla(i,s,j,q));
```

```
E_plc # Price of domestic composite, User 1 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
(TINY+PVAL1T(i,"domestic",j,q))*plc(i,j,q)
=sum(s,REGSOURCE,PVAL1A(i,s,j,q)*pla(i,s,j,q));
```

Demand for intermediate inputs, other costs and prices

E_x1c # Demand for domestic composite, User 1 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
x1c(i,j,q)=x1o(i,j,q)-SIGMA10(i)*(p1c(i,j,q)-p1o(i,j,q));

E_x1o # Demand for dom./for. composite inputs, User 1 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
x1o(i,j,q)=z(j,q)+a1(j,q);

E_xloct # Industry demand for other cost tickets #
(all,j,IND) (all,q,REGDEST)
xloct(j,q)=z(j,q)+al(j,q)+aloct(j,q);

E_ploct # Indexing of prices of other cost tickets #
(all,j,IND) (all,q,REGDEST)
ploct(j,q)=xi3(q)+floct(j,q);

```
E_efflab # Industry demand for effective labor #
(all,j,IND) (all,q,REGDEST)
efflab(j,q)=MRL(j,q)*x1prim(j,q)+a1lab(j,q)
-SIGMA1FAC(j,q)*[p1lab(j,q)+a1lab(j,q)-xi_fac(j,q)];
E_curcap # Industry demand for capital #
(all,j,IND) (all,q,REGDEST)
curcap(j,q)=MRK(j,q)*x1prim(j,q)+a1cap(j,q)
-SIGMA1FAC(j,q)*[p1cap(j,q)+a1cap(j,q)-
xi_fac(j,q)]+IL2(j,q)*interest;
E_n # Industry demand for land #
(all,j,IND) (all,q,REGDEST)
n(j,q)=MRN(j,q)*x1prim(j,q)+a1land(j,q)
-SIGMA1FAC(j,q)*[p1land(j,q)+a1land(j,q)-xi_fac(j,q)];
```

```
E_xi_fac # Effective price term for factor demand equations #
(all,j,IND) (all,q,REGDEST)
(TINY+TOTFACIND(j,q))*xi_fac(j,q)=LABOR(j,q)*(pllab(j,q)+allab(j,q))
+CAPITAL(j,q)*(plcap(j,q)+alcap(j,q))+LAND(j,q)*(plland(j,q)+alland(j,q));
```

```
E_x1laboi # Demand for labor by industry and skill group #
(all,m,OCC) (all,j,IND) (all,q,REGDEST)
x1laboi(j,q,m)=efflab(j,q)-SIGMA1LAB(j,q)*[p1laboi(j,q,m)-
p1lab(j,q)]
+IL(m,j,q)*interest;
```

```
E_pllab # Price to each industry of labor in general #
(all,j,IND) (all,q,REGDEST)
(TINY+LABOR(j,q))*pllab(j,q)=sum(m,OCC,LAB_OCC_IND(m,j,q)*pllaboi(j,q,m));
```

```
E_labind # Employment by industry #
(all,j,IND) (all,q,REGDEST)
(TINY+LABOR(j,q)) *labind(j,q)=sum(m,OCC,LAB_OCC_IND(m,j,q)*x1laboi(j,q,m));
```

```
E_x1prim # Demand for the primary-factor composite #
(all,j,IND) (all,q,REGDEST)
x1prim(j,q)=MRP(j,q)*z(j,q)+a1(j,q)+a1prim(j,q);
```

Building blocks

Producer's demands for inputs

Investor demands

- Household demands
- Export demands
- Government demands
- Margins demands
- ✓ Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Investment demand



```
E_x2a # Demand for goods by source, User 2 #
(all,i,COM) (all,s,ALLSOURCE) (all,j,IND) (all,q,REGDEST)
x2a(i,s,j,q)=IS_DOM(s)*(x2c(i,j,q)-SIGMA2C(i)*(p2a(i,s,j,q)-
p2c(i,j,q)))
+IS_IMP(s)*(x2o(i,j,q)-SIGMA2O(i)*(p2a(i,"foreign",j,q)-
p2o(i,j,q)));
```

E_p2o # Price of domestic/foreign composite, User 2 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
(TINY+PVAL20(i,j,q))*p2o(i,j,q)=sum(s,ALLSOURCE,PVAL2A(i,s,j,q)
*p2a(i,s,j,q));

```
E_p2c # Price of domestic composite, User 2 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
(TINY+PVAL2T(i,"domestic",j,q))*p2c(i,j,q)
=sum(s,REGSOURCE,PVAL2A(i,s,j,q)*p2a(i,s,j,q));
```

E_x2c # Demand for domestic composite, User 2 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
x2c(i,j,q)=x2o(i,j,q)-SIGMA2O(i)*(p2c(i,j,q)-p2o(i,j,q));

E_x20 # Demands for domestic/foreign composite, User 2 #
(all,i,COM) (all,j,IND) (all,q,REGDEST)
x20(i,j,q)=y(j,q)+a2ind(j,q);

Building blocks

- Producer's demands for inputs
- Investor demands

Household demands

- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Household demand



Each regional household determines optimal consumption bundle by maximizing a Stone-Geary utility function subject to a budget constraint

A Keynesian consumption function determines aggregate regional household expenditure

E_x3o # Household demand for composite commodities #
(all,i,COM) (all,q,REGDEST)
x3o(i,q)=[1-ALPHA_I(i,q)]*[qhous(q)+a3sub(i,q)]
+ALPHA_I(i,q)*[luxexp(q)+a3lux(i,q)-p3o(i,q)];

E_a3lux # Default setting for luxury taste shifter #
(all,i,COM) (all,q,REGDEST)
a3lux(i,q)=a3sub(i,q)-sum(k,COM,DELTA(k,q)*a3sub(k,q));

E_a3sub # Default setting for subsistence taste shifter #
(all,i,COM) (all,q,REGDEST)
a3sub(i,q)=a3com(i,q)-sum(k,COM,S3COM(k,q)*a3com(k,q));

E_utility # Change in utility disregarding taste change terms #
(all,q,REGDEST)
utility(q)=luxexp(q)-qhous(q)-sum(i,COM,DELTA(i,q)*p3o(i,q));

```
E_x3a # Demand for goods by source, User 3 #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST)
x3a(i,s,q)=IS_DOM(s)*(x3c(i,q)-SIGMA3C(i)*(p3a(i,s,q)-p3c(i,q)))
+IS_IMP(s)*(x3o(i,q)-SIGMA3O(i)*(p3a(i,"foreign",q)-p3o(i,q)));
```

```
E_p30 # Price of domestic/foreign composite, User 3 #
(all,i,COM) (all,q,REGDEST)
(TINY+PVAL30(i,q))*p30(i,q)=sum(s,ALLSOURCE,PVAL3A(i,s,q)*p3a(i,s,q));
```

```
E_p3c # Price of domestic composite, User 3 #
(all,i,COM) (all,q,REGDEST)
(TINY+PVAL3T(i,"domestic",q))*p3c(i,q)
=sum(s,REGSOURCE,PVAL3A(i,s,q)*p3a(i,s,q));
```

```
E_x3c # Demand for domestic composite, User 3 #
(all,i,COM) (all,q,REGDEST)
x3c(i,q)=x3o(i,q)-SIGMA3O(i)*(p3c(i,q)-p3o(i,q));
```

Building blocks

Producer's demands for inputs

- Investor demands
- Household demands

Export demands

- Government demands
- Margins demands
- ✓ Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Foreign export demand



Export commodities face individual downward-sloping foreign export demand functions

Exports of product i from source s are distinguished from exports of i from source r (r not equal s)

```
E_x4r # Export demand functions #
(all,i,TEXP) (all,s,REGSOURCE)
x4r(i,s)-feq(i)=EXP_ELAST(i)*[p4r(i,s)-fep(i)-natfep];
```

```
E_aggnt_x4r # Export demand functions, non-trad aggregate #
(all,s,REGSOURCE)
aggnt_x4r(s)-aggnt_feq(s)=EXP_ELAST("ADP")*[aggnt_p4r(s)-
aggnt_fep(s)-natfep];
```

```
E_nt_x4r # Export demand functions, non-trad #
(all,i,NTEXP) (all,s,REGSOURCE)
x4r(i,s)=aggnt_x4r(s)+faggnt_i(i)+faggnt_s(s)+faggnt_is(i,s);
```

```
E_aggnt_p4r # Export price, non-trad aggregate #
(all,s,REGSOURCE)
AGGEXPNT(s)*aggnt_p4r(s)=sum(i,NTEXP,PVAL4r(i,s)*p4r(i,s))+fagg
nt_p4r(s);
```

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- ✓ Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Recognise Department/Bogotá government and Central government demands for goods and services for current consumption

Neither modelled explicitly

Default:

- aggregate regional government demand in region q moves with regional private consumption, with structure of demand exogenous
- aggregate Central government demand in region q moves with national private consumption, with structure of demand exogenous

E_x5a # Regional government demand #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST)
x5a(i,s,q)=cr(q)+f5a(i,s,q)+f5gen(q)+natf5gen;

E_x6a # Central government demand #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST)
x6a(i,s,q) =natcr+f6a(i,s,q)+f6gen(q)+natf6gen;

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands

Margins demands

- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Margins commodities (identified in the set MARGCOM) provide freight services

- these commodities are consumed directly and used indirectly to facilitate the movements of products
- latter type of use is margins demand

Margins demand for margin commodity i is assumed to be proportional to the volume of the underlying flow

 e.g., margins use of transportation services in taking agricultural products to manufacturing is modelled as proportional to the volume of agricultural product used in manufacturing E_x1marg # Margins on sales to producers #
(all,i,COM) (all,j,IND) (all,q,REGDEST) (all,s,ALLSOURCE) (all,r,MARGCOM)
x1marg(i,s,j,q,r)=THETA(i,s,q)*x1a(i,s,j,q)+a1marg_ij(s,q,r)+amarg_i(s,q,r);

E_x2marg # Margins on sales to capital creators #
(all,i,COM) (all,j,IND) (all,q,REGDEST) (all,s,ALLSOURCE) (all,r,MARGCOM)
x2marg(i,s,j,q,r)=THETA(i,s,q)*x2a(i,s,j,q)+a2marg_ij(s,q,r)+amarg_i(s,q,r);

E_x3marg # Margins on sales to household consumption #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST) (all,r,MARGCOM)
x3marg(i,s,q,r)=THETA(i,s,q)*x3a(i,s,q)+a3marg_i(s,q,r)+amarg_i(s,q,r);

```
E_x4marg # Margins on exports: factory gate to port #
(all,i,COM) (all,r,MARGCOM) (all,s,REGSOURCE)
x4marg(i,s,r)=x4r(i,s)+a4marg_i(s,r);
```
Done

All demand equations

To do

- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Critical assumptions

- no pure profits in the production or distribution of commodities
- price received by the producer is uniform across all customers

Zero pure profits in current production imposed by setting unit prices received by producers equal to unit costs

Zero pure profits in distribution imposed by setting the prices paid by users equal to producer price plus commodity tax plus margins

```
E_p0a # Zero pure profits in current production #
(all,j,IND) (all,q,REGDEST)
(TINY+COSTS(j,q))*{p0a(j,q)-a(j,q)}=
sum(i,COM,sum(s,ALLSOURCE,PVAL1A(i,s,j,q)*p1a(i,s,j,q)))
+sum(m,OCC,LAB_OCC_IND(m,j,q)*p1laboi(j,q,m))
+CAPITAL(j,q)*p1cap(j,q)+LAND(j,q)*p1land(j,q)
+OTHCOST(j,q)*p1oct(j,q);
```

```
E_pi # Zero pure profits in capital creation #
(all,j,IND) (all,q,REGDEST)
(TINY+INVEST(j,q))*(pi(j,q)-a2ind(j,q))=
sum(i,COM,sum(s,ALLSOURCE,PVAL2A(i,s,j,q)*p2a(i,s,j,q)));
```

```
E_pOab # Zero pure profits in importing #
(all,i,COM)
pOa(i,"foreign")=pm(i)+natphi+powtaxm(i);
```

```
E_pla # Purchasers prices - User 1 #
(all,i,COM) (all,j,IND) (all,q,REGDEST) (all,s,ALLSOURCE)
(TINY+PVAL1A(i,s,j,q))*pla(i,s,j,q)
=[BAS1(i,s,j,q)+TAX1(i,s,j,q)]*p0a(i,s)
+BAS1(i,s,j,q)*deltax1(i,s,j,q)+sum(r,MARGCOM,MAR1(i,s,j,q,r)*
(p0a(r,q)+almarg_ij(s,q,r)+amarg_i(s,q,r)));
```

```
E_p2a # Purchasers prices - User 2 #
(all,i,COM) (all,j,IND) (all,q,REGDEST) (all,s,ALLSOURCE)
(TINY+PVAL2A(i,s,j,q))*p2a(i,s,j,q)=[BAS2(i,s,j,q)+TAX2(i,s,j,q)]*p0a
(i,s)
+BAS2(i,s,j,q)*deltax2(i,s,j,q)+sum(r,MARGCOM,MAR2(i,s,j,q,r)*
(p0a(r,q)+a2marg_ij(s,q,r)+amarg_i(s,q,r)));
```

```
E_p3a # Purchasers prices - User 3 #
(all,i,COM) (all,q,REGDEST) (all,s,ALLSOURCE)
(TINY+PVAL3A(i,s,q))*p3a(i,s,q)=[BAS3(i,s,q)+TAX3(i,s,q)]*p0a(i,s)
+BAS3(i,s,q)*deltax3(i,s,q)+sum(r,MARGCOM,MAR3(i,s,q,r)*
(p0a(r,q)+a3marg_i(s,q,r)+amarg_i(s,q,r)));
```

```
E_p4r # Purchasers prices - User 4 #
(all,i,COM) (all,s,REGSOURCE)
(TINY+PVAL4R(i,s)) * (natphi+p4r(i,s)) = [BAS4(i,s)+TAX4(i,s)]*p0a(i,s)
+BAS4(i,s)*deltax4(i,s)+sum(r,MARGCOM,MAR4(i,s,r)*
(p0a(r,s)+a4marg i(s,r)));
```

```
E_p5a # Purchasers prices - User 5 #
(all,i,COM) (all,q,REGDEST) (all,s,ALLSOURCE)
(TINY+PVAL5A(i,s,q))*p5a(i,s,q)=[BAS5(i,s,q)+TAX5(i,s,q)]*p0a(i,s)
+BAS5(i,s,q)*deltax5(i,s,q)+sum(r,MARGCOM,MAR5(i,s,q,r)*
(p0a(r,q)+a5marg_i(s,q,r)+amarg_i(s,q,r)));
```

```
E_p6a # Purchasers prices - User 6 #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST)
(TINY+PVAL6A(i,s,q))*p6a(i,s,q)=[BAS6(i,s,q)+TAX6(i,s,q)]*p0a(i,s)
+BAS6(i,s,q)*deltax6(i,s,q)+sum(r,MARGCOM,MAR6(i,s,q,r)*
(p0a(r,s)+a6marg_i(s,q,r)+amarg_i(s,q,r)));
```

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits

Indirect tax equations

- Market-clearing
- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Equations have been added to enable flexible handling of indirect taxes on all flows of goods and services

Equations allow for variations in tax rates across commodities, their sources and destinations

E_deltax1 # Tax rate on sales to User 1 #
(all,i,COM) (all,s,ALLSOURCE) (all,j,IND) (all,q,REGDEST)
deltax1(i,s,j,q)=deltax(i)+deltax1all+deltaxsource(s)+deltaxdest(q);

E_deltax2 # Tax rate on sales to User 2 #
(all,i,COM) (all,s,ALLSOURCE) (all,j,IND) (all,q,REGDEST)
deltax2(i,s,j,q)=deltax(i)+deltax2all+deltaxsource(s)+deltaxdest(q);

E_deltax3 # Tax rate on sales to User 3 #
(all,i,COM) (all,s,ALLSOURCE) (all,q,REGDEST)
deltax3(i,s,q)=deltax(i)+deltax3all+deltaxsource(s)+deltaxdest(q);

```
E_deltax4 # Tax rate on sales to User 4 #
(all,i,COM) (all,s,REGSOURCE)
deltax4(i,s)=deltax(i)+deltax4all+deltaxsource(s)
+deltaxdest("foreign");
```

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations

Market-clearing

- Regional and national macroeconomic variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Equations that impose market clearing (demand equals supply) for:

- domestically produced margin and non-margin commodities
- imported commodities

E_mkt_clear_margins # Demand equals supply for margin commodities #
(all,r,MARGCOM) (all,s,REGSOURCE)
(TINY+SALES(r,s))*z(r,s)=sum(j,IND,sum(q,REGDEST,BAS1(r,s,j,q)*x1a(r,s,j,q)
+BAS2(r,s,j,q)*x2a(r,s,j,q)))+sum(q,REGDEST,BAS3(r,s,q)*x3a(r,s,q))
+BAS4(r,s)*x4r(r,s)+sum(q,REGDEST,BAS5(r,s,q)*x5a(r,s,q))
+sum(q,REGDEST,BAS6(r,s,q)*x6a(r,s,q))
+sum(j,IND,sum(i,COM,sum(ss,ALLSOURCE,MAR1(i,ss,j,s,r)*x1marg(i,ss,j,s,r)))
+sum(i,COM, sum(ss,ALLSOURCE,MAR3(i,ss,s,r)*x3marg(i,ss,s,r))))
+sum(i,COM,sum(ss,ALLSOURCE,MAR3(i,ss,s,r)*x5marg(i,ss,s,r))))
+sum(i,COM,sum(ss,ALLSOURCE,MAR5(i,ss,s,r)*x5marg(i,ss,s,r))))
+sum(i,COM,sum(ss,ALLSOURCE,MAR5(i,ss,s,r)*x5marg(i,ss,s,r))));

Market-clearing

```
E_mkt_clear_nomarg # Demand equals supply for non-margin commodities #
(all,r,NONMARGCOM) (all,s,REGSOURCE)
(TINY+SALES(r,s))*z(r,s)=sum(j,IND,sum(q,REGDEST,BAS1(r,s,j,q)*x1a(r,s,j,q)))
+sum(j,IND,sum(q,REGDEST,BAS2(r,s,j,q)*x2a(r,s,j,q)))
+sum(q,REGDEST,BAS3(r,s,q)*x3a(r,s,q))+BAS4(r,s)*x4r(r,s)
+sum(q,REGDEST,BAS5(r,s,q)*x5a(r,s,q))+sum(q,REGDEST,BAS6(r,s,q)*x6a(r,s,q));
```

```
E_x0impa # Import volume of commodities by region #
(all,i,COM) (all,q,REGDEST)
(TINY+IMPORTS(i,q))*x0imp(i,q)=
sum(j,IND,BAS1(i,"foreign",j,q)*x1a(i,"foreign",j,q)
+BAS2(i,"foreign",j,q)*x2a(i,"foreign",j,q))
+BAS3(i,"foreign",q)*x3a(i,"foreign",q)
+BAS5(i,"foreign",q)*x5a(i,"foreign",q)+BAS6(i,"foreign",g)*x6a(i,"foreign",g);
```

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- ✓ Zero pure profits
- Indirect tax equations
- Market-clearing

✓ Regional and national macro variables and price indexes

- Government finances
- Capital accumulation and investment
- Regional population and labor market

Wide range of national and regional macro variables defined...

Two concepts of the real wage rate

- consumer real wage rate (PLAB/CPI)
- producer real wage rate (PLAB/PGDP)

Progress so far...

Done

 Core CGE equations relating to demand, supply, prices, indirect taxes, market-clearing and summary macro variables

To do

- Government finances
- Capital accumulation and investment
- Regional population and labor market

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations
- ✓ Market-clearing
- Regional and national macro variables and price indexes

Government finances

- Capital accumulation and investment
- Regional population and labor market

Value added disaggregation

Gross regional product

Miscellaneous linking equations

Summary of financial transactions

Household income

Government finances



KEY

- Government finance block of equations.
 - The five equation groupings within the government finance block.
 - Other equation blocks that link with government finances.
- \rightarrow Direction of link between equation groupings, and other equation blocks.

All equations blocks have backward links to the CGE core

- expenditure on primary factors is disaggregated into gross returns to primary factors and production taxes
- gross returns to primary factors is used in the determination of the income side of gross regional product
- factor incomes are used to determine household income
- production taxes are used in the determination of government budget balance



- **z_**_r** value-added component in gross regional production
- **dompy***** domestic regional production, income component
- **dompq***** domestic regional production, expenditure component
- **softy***** summary of financial transaction, income component
- **softq***** summary of financial transaction, expenditure component
- **hhidy***** household income component

Disaggregation of regional value added



KEY

* The name in parenthesis indicates the corresponding array name in Figure 2.1

Elements from the CGE Core.

Elements in the disaggregation-of-value-added

Core variables for industry wage bill, capital rental, land rental and other cost tickets disaggregated into:

- wages, salaries and supplements, imputed wages, payroll tax
- returns to fixed capital, property taxes
- returns to land, land taxes
- returns to working capital, other indirect taxes, sales by final buyers

Disaggregated components added to national account concepts of Wage income, GOS and Production taxes.

Together these comprise value added

Allowance also made for "Sales by final buyers"

Gross Regional Product



Value added plus indirect taxes net of subsidies add to Gross Regional Product from the income side

Indirect taxes are the sum of tariffs and other commodities taxes net of subsidies taken from the CGE core

From the **expenditure side**, Gross Regional Product is the sum of expenditure by final users (government, households, investors and foreigners) less international imports plus net trade with the rest of Colombia

Gross Regional Product



Summary of financial transactions: SOFT accounts



Household disposable income



Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- ✓ Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macro variables and price indexes
- Government finances
- Capital accumulation and investment
- Regional population and labor market

Capital, investment and expected rates of return

$$K_{j,q}(t+1) = (1 - DEP_{j,q}) \times K_{j,q}(t) + Y_{j,q}(t)$$

Given starting point for capital (t=0) and an explanation of investment, we can trace out time path for capital

Investment explained by assuming that:

$$\frac{K_{j,q}(t+1)}{K_{j,q}(t)} - 1 = F_{j,q}^{t} [EROR_{j,q}(t)]$$

Growth in capital related to expected rate of return

 In CEER only assume static expectations, though rational is possible For static expectations case, the actual rate of return is:

$$RO_t(j,q) = \frac{P_t(j,q)}{\prod_t(j,q)} - D(j,q)$$

$$ro(j,q) = p_t(j,q) - \pi_t(j,q)$$

 $ro(j,q) = QCOEF(j,q) [p_t(j,q) - \pi_t(j,q)]$

QCOEF: relação entre taxa bruta e taxa líquida de retorno (> 1)

In long-run comparative-static simulations:

- aggregate capital adjusts to maintain R_{INT} (natr_tot)
- capital allocated in line with equation **E_f_rate_xx**
 - industries with relatively large increases in capital require relatively high rates of return
 - industries with relatively small increases in capital require relatively low rates of return
- industry investment determined by fixed ratios of investment to capital (equation **E_y**)

Equalization in the rates of return

$$\left(\frac{K(j,q)}{K(q)}\right)^{-\beta(j,q)} RO(j,q) = R_{\text{int}}$$

$$ro(j,q) - r_{int} = \beta_t(j,q) [k(j,q) - k(q)] + f _ rate(j,q)$$

beta: risk/return ratio

Short-run: *f_rate* endogenous, *k* exogenous

Long-run: *f_rate* exogenous, *k* endogenous

Growth rate of capital stocks and investment in the short-run:

 $k_{t+1}(j,q) - k_t(j,q) = 0$

% change in capital stocks

 $y_t(j,q) = 0$

% change in investment

Growth rate of capital stocks and investment in the long-run:

$$\frac{K_{j,q}(t+1)}{K_{j,q}(t)} = \left(\frac{K_{j,q}(t)}{K_{j,q}(0)}\right)^{1/T}$$

$$k_{t+1}(j,q) = \left(1 + \frac{1}{T}\right)k_t(j,q)$$
Fixed capital stocks in the base year values:

- curcap(j,q) exogenous (=0)
- relationship between sectoral rates of return,, r0(j,q), and reference interest rate, natr_tot, is endogenous (f_rate_xx(j,q) endogenous)

Percentage change in sectoral investment, y(j,q) is zero; this can be guaranteed by setting the shift term, $delf_rate(j,q)$, exogenous and zero

By hypothesis, not only the capital stocks are fixed but also firms' investment plans

Investment in the short run

E_r0 # Definition of rates of return to capital # r0(j,q) = QCOEF(j,q)*(plcap(j,q) - pi(j,q));

E_f_rate_xx # Capital growth rates related to rates of return # (r0(j,q) - natr_tot) = BETA_R(j,q)*[curcap(j,q) kt(q)] + f_rate_xx(j,q);

E_curcapT1 # Capital stock in period T+1 # curcap_t1(j,q) - curcap(j,q) =0;

E_yT # Investment in period T # curcap(j,q) - y(j,q) - 100*delf_rate(j,q)=0;



Capital stocks endogenously determined:

- curcap(j,q) endogenous
- relationship between sectoral rates of return,, r0(j,q), and reference interest rate, natr_tot, is given (f_rate_xx(j,q) exogenous)

Percentage change in sectoral investment, y(j,q) is endogenous

Firms' investment plans are carried out, reestablishing returns differentials in the base year

Rate of capital accumulation, but **not the level** of capital stock, remains constant

E_r0 # Definition of rates of return to capital # r0(j,q) = QCOEF(j,q)*(plcap(j,q) - pi(j,q));

E_f_rate_xx # Capital growth rates related to rates of return # (r0(j,q) - natr_tot) = BETA_R(j,q)*[curcap(j,q) - kt(q)] + f_rate_xx(j,q);

E_curcapT1 # Capital stock in period T+1 # curcap_t1(j,q) - K_TERM*curcap(j,q)=0;

endog. exog.

E_yT # Investment in period T # VALK_T1(j,q)*curcap_t1(j,q) = VALKT(j,q)*DEP(j)*curcap(j,q) +(INVEST(j,q))*y(j,q)-100*(VALK_0(j,q)*(1-DEP(j))

(CEER adopts DEP(j) = 0.96)

Fipe - Fundação Instituto de Pesquisas Econômicas 76

Building blocks

- Producer's demands for inputs
- Investor demands
- Household demands
- Export demands
- Government demands
- Margins demands
- Zero pure profits
- Indirect tax equations
- Market-clearing
- Regional and national macro variables and price indexes
- Government finances
- Capital accumulation and investment

Regional population and labor market

Regional population and labor market

Critical variables:

- regional population
- regional migration
- regional unemployment
- regional participation rates
- regional wage relativities

Various closures

Regional population and labor market

(1) Fixed

 wage relativities (determining employment by region), participation and unemployment rates (determining population by region)

(1) Endogenous

regional migration

(2) Fixed

regional migration, participation rates, wage relativities

(2) Endogenous

unemployment rates

- (3) Fixed
 - regional migration, participation and unemployment rates
- (3) Endogenous
 - wage relativities

del_unr(q) # Percentage-point changes in regional unemployment rate #;

endog. exog.

E_wage_diff # Region real-wage diff #(all,q,REGDEST)
wage_diff(q) = pwage(q) - natxi3 - natrealwage;
E_del_labsup # P-point changes in regional
unemployment rates #(all,q,REGDEST)
C_labsup(q)*del_unr(q)=C_EMPLOY(q)*(labsup(q)employ(q));

del_unr(q) # Percentage-point changes in regional unemployment rate #;

🗕 endog. 🛛 💻 exog.

Fipe - Fundação Instituto de Pesquisas Econômicas 81



Each equation explains a variable

More variables than equations

Endogenous variables: explained by model

Exogenous variables: set by user

Closure: choice of exogenous variables

Many possible closures

Number of endogenous variables = Number of equations

T is related to our choice of closure

With short-run closure we assume that:

- T is long enough for price changes to be transmitted throughout the economy, and for price-induced substitution to take place
- T is not long enough for investment decisions to greatly affect the useful size of sectoral capital stocks [new buildings and equipment take time to produce and install]

T might be 2 years. So results mean:

 A 10% consumption increase might lead to employment in 2 years time being 1.2% higher than it would be (in 2 years time) if the consumption increase did not occur. Many closures might be used for different purposes

No unique natural or correct closure

Must be at least one exogenous variable measured in local currency units

Normally just one — called the *numéraire*

Often the exchange rate, *natphi*, or *natxi3*, the CPI.

Some quantity variables must be exogenous, such as:

- primary factor endowments
- final demand aggregates

Short-run environment



Long-run environment

